Maple-assisted proof of empirical formula for A267241

Robert Israel

Sep 8 2019

Consider a "state" of the system to be a 1×7 binary array x, where $x_1 .. x_4$ represent a row, $x_5 = 1$ if this row and the previous ones have already determined that column 2 is lexicographically greater than column 1, and similarly for x_6 with column 3 and column 2 and x_7 with column 4 and column 3. In particular if $x_1 \neq x_2$ we must have $x_5 = 1$ and similarly for the others.

We enumerate the 54 possible states:

```
> states:= select(proc(x) (x[1]=x[2] or x[5]=1) and (x[2]=x[3] or x
  [6]=1) and (x[3]=x[4] or x[7]=1) end proc, [seq(seq(seq(seq(seq
  (seq(seq([a,b,c,d,e,f,g],g=0..1),f=0..1),e=0..1),d=0..1),c=0..1),
  b=0..1),a=0..1)]): nops(states);
```

$$54 \quad (1)$$

Although it is not part of the $n \times 4$ array, we may imagine that we start in state [0,0,0,0,0,0,0]. Let T be the 54×54 transition matrix where $T_{ij} = 1$ if state j can be followed by state i.

```
> T:= Matrix(54,54,proc(i,j) local k;
    if add(states[j,k]-states[i,k],k=1..4) > 0 then return 0 fi;
    if states[j,5]>states[i,5] or states[j,6]>states[i,6] or states
  [j,7]>states[i,7] then return 0 fi;
    if states[i,1]>=states[i,2] and states[j,5]<> states[i,5] then
  return 0 fi;
    if states[i,2]>=states[i,3] and states[j,6]<> states[i,6] then
  return 0 fi;
    if states[i,3]>=states[i,4] and states[j,7]<> states[i,7] then
  return 0 fi;
  1
  end proc):
```

Then we should have $a_n = u^T T^n e$ where $u = (1, ..., 1)^T$ and $e = (1, 0, ..., 0)^T$. To check, we compute the first few terms of the sequence. .

```
> U:= Vector(54,1):
  E[0]:= Vector(54): E[0][1]:= 1:
  for k from 1 to 25 do E[k]:= T . E[k-1] od:
  seq(U^%T . E[j], j=1..25);
```

5, 22, 105, 567, 3351, 20676, 129129, 804817, 4982759, 30629206, 187121865, 1137631979, 6891047527, 41628865000, 250987078681, 1511105743781, 9088662549303, 54625229882746, 328144877989145, 1970524978549951, 11830099105261335, 71009696059657932, 426179797614950025, 2557586127460436217, 15347629546703286471 **(2)**

Now the empirical formula is

```
> Emp:=   a(n) = 24*a(n-1) -246*a(n-2) +1420*a(n-3) -5121*a(n-4)
  +12084*a(n-5) -18944*a(n-6) +19536*a(n-7) -12720*a(n-8) +4736*a
  (n-9) -768*a(n-10):
```

This corresponds to $u^T P(T) T^n e = 0$ where $P(x)$ is the following polynomial of degree 10:

```
> P:= x^10 - add(coeff(rhs(Emp),a(n-i))*x^(10-i),i=1..10);
```

$$P := x^{10} - 24 x^9 + 246 x^8 - 1420 x^7 + 5121 x^6 - 12084 x^5 + 18944 x^4 - 19536 x^3 + 12720 x^2 - 4736 x + 768 \quad (3)$$

It turns out that $u^T P(T) = 0$. This completes the proof.

```
> U^%T . eval(P,x=T);
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0] **(4)**